
ВВЕДЕНИЕ

Настоящие методические указания предназначены для обеспечения

учебного процесса студентов четвертого курса дневной формы обуче-
ния специальности 220200 “АСОИиУ” при выполнении лабораторной
работы по предмету “Методы искусственного интеллекта, базы зна-
ний, экспертные системы”.

Цели лабораторной работы: изучить возможности средств языка
программирования Пролог для создания экспертных систем и обра-
ботки реляционных данных, в частности, описывающих электрические
схемы.

1. СИСТЕМНЫЕ ТРЕБОВАНИЯ

Для выполнения лабораторной работы необходимы следующие си-
стемные компоненты:
а) компьютер, работающий под управлением операционной системы

Linux;
б) транслятор Пролога, соответствующий стандарту ISO/IEC 13211-

1, например, GNU Prolog.

2. ПОДГОТОВКА ИСХОДНЫХ ДАННЫХ

Данные, описывающие электрическую схему, естественно предста-
вить в виде таблицы. Например, электрическую схему на рис. 1

Рис. 1

естественно представить в виде следующей таблицы:

1-й узел 2-й узел сопротивление

1 2 R1

2 3 R2

1 4 R3

2 5 R4

4 5 R5

5 3 R6 .

Очевидно, что подобной таблицей можно описать любую, сколь
угодно сложную схему.

3. ОБРАБОТКА ТАБЛИЦ СРЕДСТВАМИ

ЯЗЫКА ПРОГРАММИРОВАНИЯ ПРОЛОГ

Данные, представимые в виде таблиц, представляют собой част-
ный случай более общих структур, называемых фразами Хорна. Язык
программирования Пролог позволяет по заданной совокупности фраз

Хорна (программе или, точнее, базе данных) автоматически получать
ответы на заданный запрос, описываемый также фразой Хорна. Язык
фраз Хорна является частью языка логики предикатов 1-го порядка.
Предикатами называют функции, которые могут возвращать толь-
ко логические значения. У предикатов 1-го порядка не может быть
аргументов-предикатов. Вследствие того, что эффективная машинная
обработка даже такой небольшой части языка логики, как язык фраз
Хорна, чисто логическими средствами невозможна, Пролог использу-
ет другие (процедурные) методы поиска логических выводов. Выводы,
получаемые Прологом, всегда являются частью всех возможных логи-
чески правильных ответов на заданный запрос. Из-за того, что поиск
решений основан на последовательном вызове подпрограмм, соответ-
ствующих предикатам базы данных и запроса, а не на логическом вы-
воде, Пролог имеет ряд средств для контроля за ходом поиска решений,
которые не имеют к логике никакого отношения.

Фразы Хорна, составляющие базу данных, делятся на два типа —
факты и правила. Первые представляют собой отдельные предикаты-
утверждения, а вторые соответствуют структуре предложения “если–
то”, часть “если” которого состоит из одного или нескольких, соединен-
ных союзом “и” предикатов-условий, а часть “то” из одного предиката-
заключения. Факты можно рассматривать как частный случай правил,
состоящих только из части “то”. Операция логическое “и” в Прологе
обозначается запятой, а операция логического следования— последова-
тельностью двух символов, двоеточия и дефиса. Следование в Прологе
записывается справа-налево, т.е. сначала заключение, а затем посылка.
Если одному заключению соответствуют разные посылки, то это озна-
чает наличие между этими посылками отношения, соответствующего
операции логическое “или”, которая обозначается символом “точка с
запятой”. Фраза-запрос состоит из одного или нескольких разделенных
запятыми (связками “и”) предикатов. Часть “то” фактов и правил на-
зывается головой фразы, а часть “если” — телом. Факты не имеют

тела, а запросы головы. Все фразы завершаются точкой.
Весь запрос называется целью, а предикаты, его составляющие, —

подцелями. Все переменные, входящие в предикаты цели, называются
целевыми. При поиске решения ключевую роль играет поиск сопоста-
вимых предикатов и их дальнейшая унификация.

Два предиката сопоставимы, если они имеют одинаковые имена и
местности и, кроме того, ни одна пара их соответствующих аргументов
не состоит из разных констант. Унификация определяется подстанов-
кой, делающей оба предиката идентичными. В ней паре соответствую-
щих различных аргументов соответствует замена либо переменной на

константу, либо переменной на переменную.
Поиск решения Прологом проходит согласно следующей последо-

вательности шагов:
1) Если цель пуста, то получено одно решение, которое либо состоит
из значений целевых переменных, либо, если переменных в запросе
нет, — из ответа “да”. Значения целевых переменных получаются
последовательным применением к ним унифицирующих подстано-
вок;

2) Вычисляется первая слева подцель в запросе. Она может состо-
ять из предопределенного предиката или из предиката, заданно-
го в базе данных. В первом случае, если вычисленное значение
предиката истинно, то происходит его отбрасывание, иначе поиск
решения для заданной цели прекращается. Во втором случае, ба-
за данных просматривается последовательно от начала до конца

(порядок фраз в базе данных имеет значение) до тех пор, пока го-
лова очередной фразы не станет сопоставимой с подцелью. Если
найденная фраза содержит переменные, встречавшиеся в целях, то
перед унификацией их надо заменить на другие, в целях не встре-
чавшиеся. После чего вычисляется унифицирующая подстановка,
подцель заменяется на тело найденной фразы и к полученной но-
вой цели применяется вычисленная унифицирующая подстановка.
Переход к шагу 1. Если же не найдется ни одной фразы с головой,
сопоставимой с подцелью, то поиск решения для заданной цели
прекращается.
Состояние, при котором дальнейший поиск решения для заданной

цели невозможен, называется “неуспехом”. Таким образом, поиск реше-
ния состоит в замене цели до тех пор, пока она либо не станет пустой
(“успех”), либо невозможной (“неуспех”). Если в запросе нет перемен-
ных, то поиск заканчивается после нахождения одного решения. Если
же переменные в запросе есть, то в случае как “успеха”, так и “не-
успеха” происходит возврат назад по цепочке целей до тех пор, пока
очередная предыдущая цель не будет допускать альтернативы при за-
мене своей первой подцели, т.е. если за найденной при поиске решения

фразой в базе данных есть еще фраза, голова которой сопоставима с
заданной подцелью. Для найденной цели-альтернативы поиск решения
также проходит согласно приведенному двухшаговому алгоритму. Ес-
ли же все альтернативы исчерпаны, то если ни разу не был достигнут
“успех”, то результат ответ “нет”, иначе ответом являются один или
более наборов значений целевых переменных. Описанный метод поиска
решений называется “поиск с возвратом”.

Если установить резисторам на рис. 1 конкретные значения, на-
пример, R1 = 100 Ом, R2 = R4 = 300 Ом, R3 = R5 = R6 = 500 Ом,
то таблицу, описывающую эту схему, можно представить следующим
далее набором фактов базы данных Пролога.

connection(100, 1, 2).

connection(300, 2, 3).

connection(500, 1, 4).

connection(300, 2, 5).

connection(500, 4, 5).

connection(500, 5, 3).

Эта база данных, как и эквивалентная ей таблица, имеет два суще-
ственных недостатка. Во-первых, она фиксирует направление от кон-
такта к контакту в каждом соединении. Информация о том, что сопро-
тивление резистора не зависит от выбранного направления между кон-
тактами, в базе данных никак не отражена. Во-вторых, на схеме могут
быть несколько идентичных соединений, которые было бы желатель-
но четко различать. С другой стороны подобная база данных наиболее
естественна для ввода в качестве исходной. Таким образом, получается,
что перед собственно расчетами нужно пополнить базу данных табли-
цей, которая будет как содержать всю исходную информацию, так и
информацию, избавляющую от двух означенных недостатков. В этой

новой таблице одному соединению будут соответствовать две строки,
задающих оба возможных направления при соединении, с идентичными
полями сопротивления и дополнительного идентифицирующего номера

соединения. Эта новая таблица должна создаваться из исходной авто-
матически в начале расчетов.

Расчет сопротивления между двумя заданными контактами на схе-
ме будет проходить через серию преобразований базы данных, цель ко-
торых свести всю схему к эквивалентному ей по сопротивлению одному

соединению. Означенные преобразования базы данных соответствуют
простейшим электротехническим эквивалентным расчетным заменам:
1) два последовательно соединенных резистора, заменяются одним с
сопротивлением, равным сумме сопротивлений исходных резисто-
ров;

2) два параллельно соединенных резистора, заменяются одним с вели-
чиной обратной сопротивлению, равной сумме величин, обратных
сопротивлениям исходных резисторов;

3) изолированная ветвь или петля (примеры таких ветви и петли,
возникающих при измерении сопротивления между контактами 1
и 3, показаны соответственно на рис. 2 и рис. 3) отбрасывается;

4) трехлучевая звезда резисторов, заменяется на треугольник соглас-
но рис. 4.

Рис. 4

Первые два преобразования сокращают базу данных на одно со-
единение каждое. Четвертое исключает из базы данных один контакт.
Третье сокращает базу данных на одно соединение, а в случае с изоли-
рованной ветвью еще и исключает один контакт.

Сначала производятся все возможные простейшие преобразования

1, 2 и 3. Если в результате не удастся получить одно соединение, то
производится 4-е преобразование и затем все возможные простейшие

упрощения 1–3. Если в результате последних получить одно соедине-
ние опять не удастся, то цикл повторяется с 4-го преобразования. Если
же когда-нибудь после 4-го преобразования не удастся выполнить ни
одного упрощения 1–3, то это будет означать неспособность данного
метода поиска решения найти ответ на запрос.

4. ФОРМАЛЬНАЯ ЛОГИКА ПОИСКА РЕШЕНИЙ И

ВСПОМОГАТЕЛЬНЫЕ ОПИСАНИЯ

Пролог не позволяет изменять фразы, заданные в тексте програм-
мы. Данные из текста программы называются поэтому статическими,
в отличие от динамических данных, которые могут произвольно ме-
няться по ходу поиска решений. Пример статических данных — это

предикат connection, пример динамических — это создаваемая по ин-
формации из него новая таблица, описывающая предикат branch.

Все динамические предикаты должны быть объявлены до их пер-
вого использования стандартным предикатом dynamic. Кроме новой
таблицы, branch, полностью описывающей схему, нужны еще будут та-
блица из двух строк и одного столбца, extra point, для хранения двух
точек измерения, а также одностолбцовая таблица, t, для временных
величин.

:- dynamic(branch/4, extra point/1, t/1).

В приведенном объявлении после имени предиката через наклон-
ную черту следует его местность. Это объявление оформлено как за-
прос. Он выполняется до ввода запроса пользователем.

Затем в текст программы заносится описание схемы для расчета

предикатом connection.
Затем следует описание метода поиска решения на языке фраз

Хорна. Поиск решения инициируется запросом, 3-местным предикатом
query, 1-й аргумент которого целевая переменная, искомое сопротивле-
ние, а два оставшихся — номера точек измерения.

После получения запроса нужно уничтожить все оставшиеся от

возможных предыдущих расчетов динамические данные и затем пра-
вильно заполнить динамические таблицы branch и extra point.

query(R,X,Y) :-

retractall(extra point( )),

assertz(extra point(X)), assertz(extra point(Y)),

fill branch; % инициализация динамических данных

simplify, branch(R,X,Y, ), !; % расчет

write(’I can’’t get solution...’), nl.

fill branch :- % заполнение branch

retractall(branch( , , , )), retractall(t( )),

assertz(t(0)), fail;

connection(R,X,Y), t(N), N1 is N+1, retract(t(N)),

assertz(t(N1)), assertz(branch(R,X,Y,N)),

assertz(branch(R,Y,X,N)), fail.

Всегда истинный предикат assertz добавляет утверждение-аргу-
мент к концу динамической базы данных, а всегда истинный предикат
retractall удаляет из динамической базы данных все фразы, голова ко-
торых сопоставима с его аргументом. Предикат retract отличается от
retractall тем, что он за один раз удаляет только одну фразу и создает
точку разветвления для поиска решений. Если retract не сможет уда-
лить ни одной фразы, то его результат — ложь. Переменные в Прологе
должны начинаться с заглавной буквы или с подчеркивания. Перемен-
ная, имя которой состоит из одного знака подчеркивания, называется
анонимной. Все анонимные переменные считаются различными и их
значения не выводятся в ответах. Всегда истинные предикаты write и
nl печатают соответственно значение своего аргумента и конец строки.
“Предикат” fail всегда ложен — это не логическое средство, для опи-
сания логики он бессмыслен. Однако, в Прологе, поставленный в кон-
це фразы, он прекратит ее вычисление только после вычисления всех
предшествующих ему предикатов. Это и связка “или” позволяют ор-
ганизовывать вспомогательные расчеты. Всегда истинный “предикат”
отсечения, обозначаемый восклицательным знаком, также не является
логическим средством — он используется из-за своего побочного эф-
фекта. После выполнения отсечения поиск решения прекращается по
целям-альтернативам, возникающим по ходу выполнения предиката,
вызвавшего предикат с отсечением. В частности, выполнение отсече-
ния в предикате означает отказ от вычисления следующих далее через

связку “или” его частей.
Предикат fill branch всегда ложен — после него через связку “или”

следует главная часть программы — попытка упрощения схемы до од-
ного эквивалентного соединения. Предикат simplify истинен, если такое
упрощение удастся сделать и ложен в противном случае. Если упро-
щение удалось, то печатается сопротивление полученного соединения;
если нет, то — сообщение о том, что программа не может найти реше-
ние.

simplify :-

simplify1;

simplify2;

calcnbr(2).

simplify1 :-

retractall(t(1)), parallel;

sequential;

cut;

t(1), simplify1.

simplify2 :-

retractall(t(2)), star, simplify1;

t(2), simplify2.

calcnbr(X) :-

retractall(t( )), assertz(t(0)), fail;

branch( , , , ), t(N), N1 is N+1,

retract(t(N)), assertz(t(N1)), fail;

t(N), X = N.

Предикат simplify1 пытается упростить схему всеми возможными
простейшими преобразованиями 1–3, осуществляемые предикатами se-
quential, parallel и cut, которые, кроме того, устанавливают истинными
t(1) и t(2) при осуществлении преобразований. Предикат simplify2 пе-
ред тем как перейти к простейшим преобразованиям выполняет преоб-
разование 4, предикатом star. И simplify1, и simplify2 — всегда ложны,
каждый из них выполняется до тех пор, пока ни одно из простейших
преобразований не сможет быть выполнено. После их выполнения про-
изводится расчет количества соединений, оставшихся после преобразо-
ваний схемы, предикатом calcnbr. Если это число равно 2, то simplify
устанавливается истинным; если нет, то — ложным.

sequential :-

branch(R1,X,Y,N1), branch(R2,Y,Z,N2), N1 \== N2, simple2(Y),

retract(branch(R1,X,Y,N1)), retract(branch(R2,Y,Z,N2)),

retract(branch(R1,Y,X,N1)), retract(branch(R2,Z,Y,N2)),

R is R1+R2,

assertz(branch(R,X,Z,N1)), assertz(branch(R,Z,X,N1)),

assertz(t(1)), assertz(t(2)), fail.

simple2(X) :-

branch( ,X, ,N1), branch( ,X, ,N2), N1 \== N2,

branch( ,X, ,N3), N1 \== N3, N2 \== N3, !, fail;

extra point(X), !, fail;

true.

parallel :-

branch(R1,X,Y,N1), branch(R2,X,Y,N2), N1 \== N2,

retract(branch(R1,X,Y,N1)), retract(branch(R2,X,Y,N2)),

retract(branch(R1,Y,X,N1)), retract(branch(R2,Y,X,N2)),

R is R1*R2/(R1+R2),

assertz(branch(R,X,Y,N1)), assertz(branch(R,Y,X,N1)),

assertz(t(1)), assertz(t(2)), fail.

cut :-

branch( ,X,X, ), retractall(branch( ,X,X, )),

assertz(t(1)), assertz(t(2)), !, fail; % удаление петель

branch( ,X, , ), simple1(X),

retractall(branch( ,X, , )), retractall(branch( , ,X, )),

assertz(t(1)), assertz(t(2)),

fail. % удаление изолированных соединений

simple1(X) :-

branch( ,X, ,N1), branch( ,X, ,N2), N1 \== N2, !, fail;

extra point(X), !, fail;

true.

star :-

branch(R1,X1,Y,N1), branch(R2,X2,Y,N2), N1 \== N2,

branch(R3,X3,Y,N3), N1 \== N3, N2 \== N3, simple3(Y),

retract(branch(R1,X1,Y,N1)), retract(branch(R1,Y,X1,N1)),

retract(branch(R2,Y,X2,N2)), retract(branch(R2,X2,Y,N2)),

retract(branch(R3,X3,Y,N3)), retract(branch(R3,Y,X3,N3)),

RT is R1*R2+R2*R3+R1*R3,

R12 is RT/R3, R23 is RT/R1, R13 is RT/R2,

assertz(branch(R12,X1,X2,N1)), assertz(branch(R12,X2,X1,N1)),

assertz(branch(R13,X1,X3,N2)), assertz(branch(R13,X3,X1,N2)),

assertz(branch(R23,X2,X3,N3)), assertz(branch(R23,X3,X2,N3)).

simple3(X) :-

branch( ,X, ,N1), branch( ,X, ,N2), N1 \== N2,

branch( ,X, ,N3), N1 \== N3, N2 \== N3,

branch( ,X, ,N4), N1 \== N4, N2 \== N4, N3 \== N4, !, fail;

extra point(X), !, fail;

true.

Предикаты simple1(X), simple2(X) и simple3(X) истины в случае,
если X — это узел на схеме, в котором пересекаются не более соответ-
ственно одного, двух и трех соединений.

5. ПОРЯДОК ВЫПОЛНЕНИЯ И ОТЧЕТ

ПО ЛАБОРАТОРНОЙ РАБОТЕ

5.1. Порядок выполнения лабораторной работы
Лабораторная работа выполняется за следующие четыре этапа:

1) внесение в базу знаний описания логики поиска решения;
2) дополнение базы знаний фактами, описывающими заданную инди-
видуальным вариантом схему;

3) в Пролог-системе ввести запросы к введенной на предыдущих эта-
пах базе знаний для получения ответов о сопротивлении заданных

вариантом задания участков;
4) проанализировать ход выполнения Пролог-программы по каждому
из запросов.
Ход преобразований при выполнении запроса можно отслеживать

добавлением к определениям упрощающих предикатов cut, parallel, se-
quential и star части, печатающей соответствующую информацию при
выполнении преобразования. Например, в описание предиката sequen-
tial, перед fail можно поставить

write(’sequential(’), write(X), write(Y), write(Z), write(’)’).

5.2. Отчет по лабораторной работе
Отчет по работе выполняется на отдельных листах или в отдель-

ной тетради и должен содержать:
1) цель исследования;
2) спецификацию задания;
3) полученные ответы и описание хода их получения Пролог-систе-
мой;

4) выводы.

6. КОНТРОЛЬНЫЕ ВОПРОСЫ

1. Назначение языка программирования Пролог, его сильные и сла-
бые стороны.

2. Связь логических и табличных данных.
3. Недостатки и преимущества использования Пролога при расчетах
характеристик сложных электрических цепей по сравнению с ме-
тодом, опирающемся на построение системы линейных уравнения
по законам Кирхгофа.

4. Пути совершенствования приведенной в настоящем указании экс-
пертной программы.

5. Логическое описание реализации методов упрощения электриче-
ской схемы, описанных в базе знаний.

6. Методы формального описания электрических схем, их достоин-
ства и недостатки.

7. Оптимальной ли была стратегия поиска решения при поиске отве-
тов на запросы?

7. ВАРИАНТЫ РАБОТ

Лабораторная работа имеет 25 вариантов заданий на основе пяти
принципиальных электрических схем, приведенных на рис. 5–9. Каж-

дый индивидуальный вариант задается номером рисунка и двумя пара-
ми номеров контактов с этих рисунков, сопротивления между которыми
нужно рассчитать. Для всех вариантов R1 = R3 = 100 Ом, R2 = R7 =
R8 = 200 Ом, R4 = R5 = R6 = 300 Ом, R9 = R12 = R14 = 400 Ом,
R10 = R11 = 700 Ом, R13 = 600 Ом.

Вариант № рисунка 1-я пара 2-я пара
1 5 3, 8 4, 5
2 5 4, 6 5, 7
3 5 1, 7 1, 6
4 5 9, 10 1, 9
5 5 2, 6 4, 8
6 6 3, 4 3, 6
7 6 1, 2 4, 5
8 6 1, 8 1, 6
9 6 3, 7 1, 7
10 6 2, 6 2, 4
11 7 1, 7 1, 6
12 7 1, 2 1, 5
13 7 5, 6 3, 9
14 7 3, 6 3, 7
15 7 4, 5 8, 9
16 8 1, 2 1, 5
17 8 1, 6 7, 8
18 8 2, 4 2, 7
19 8 3, 9 1, 3
20 8 3, 4 3, 5
21 9 1, 6 1, 7
22 9 1, 5 6, 11
23 9 2, 8 2, 9
24 9 5, 11 1, 8
25 9 4, 6 4, 5

